

```

' I2CREAD and I2WRITE TEST ONE
'
' By Squibcakes 17/03/2006
'
' Write to the first 16 locations of an external serial EEPROM
' Read first 16 locations back and send to serial output
' Note: for EEPROMS with word-sized address
'       (ie 24LC256 32k Eeprom with 15 bit addressing 0000H:8000H, 0->32767 DEC
)
'       : SCK connected via 2K pulled high
'       : SDI connected via 2K pulled high
'       : A0:A2 connected to ground
'       : WP connected to ground

' =====
' DEVICE PROGRAMMING OPTIONS
' =====

@ DEVICE PIC16F88, WDT_ON
@ DEVICE PIC16F88, HS_OSC
@ DEVICE PIC16F88, BOD_ON
@ DEVICE PIC16F88, PROTECT_OFF
@ DEVICE PIC16F88, MCLR_ON
@ DEVICE PIC16F88, CCPMX_OFF

DEFINE LOADER_USED 1      ' Here we are using Tiny Bootloader to load program
' =====
' SET USART PARAMS
' =====

DEFINE HSER_RCSTA 90h      ' Set receive register to receiver enabled
DEFINE HSER_TXSTA 20h      ' Set transmit register to transmitter enabled
DEFINE HSER_BAUD 4800      ' Set baud rate
DEFINE HSER_SPBRG 64      ' Set SPBRG directly (normally set by HSER_BAUD)

' =====
' SET OSCILLATOR
' =====
DEFINE OSC 20

' =====
' CONFIGURE COMPARATOR MODULE
' =====
CMCON = 7                  ' TURN ANALOG COMPARATOR MODE OFF

' =====
' PROGRAM VARS
' =====
CPIN   VAR    PORTB.4      ' I2C clock pin
DPIN   VAR    PORTB.1      ' I2C data pin
CTRL   VAR    BYTE         ' Control byte
ADDRESS VAR    WORD        ' Eeprom addressing word 15 bit address
B1     VAR    BYTE
B2     VAR    BYTE

' Control byte value

CTRL = %10100000

' =====
' WRITE TO EEPROM
' =====

HSEROUT["Write to I2C.",13,10,"Press space to continue",13,10]
HSERIN [WAIT(" ")]

FOR ADDRESS = 0 TO 15      ' Loop 16 times
    B1 = ADDRESS.LowByte      ' get address value and write

```

```
        I2CWRITE DPIN,CPIN,CTRL,ADDRESS,[B1]    ' back to itself.
        PAUSE 10                                ' Delay 10ms after each write
        HSEROUT [#B1," "]

    NEXT ADDRESS

    PAUSE 1000

' =====
' READ BACK EEPROM
' =====

    HSEROUT[13,10,"Read from I2C.",13,10,"Press space to continue",13,10]

    HSERIN [WAIT(" ")]

    FOR ADDRESS = 0 TO 15 STEP 2
        I2CREAD DPIN,CPIN,CTRL,ADDRESS,[B1,B2] ' Read data location
        HSEROUT [#B1," ",#B2," "]             ' Send data to Serial Port
    NEXT ADDRESS

    HSEROUT[13,10,"I2C Write and Read Test Completed."]

    STOP
    END
```